# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-bdftopcf
pkgver=1.0.5
pkgrel=1
pkgdesc="Convert X font from Bitmap Distribution Format to Portable Compiled Format"
arch=(i686 x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxfont')
makedepends=('xorg-util-macros' 'xproto')
groups=('xorg' 'xorg-apps')
source=(https://xorg.freedesktop.org/archive/individual/app/bdftopcf-${pkgver}.tar.bz2{,.sig})
sha256sums=('38f447be0c61f94c473f128cf519dd0cff63b5d7775240a2e895a183a61e2026'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd "${srcdir}/bdftopcf-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/bdftopcf-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
